

МИНИСТЕРСТВО ОБРАЗОВАНИЯ И НАУКИ РФ  
ФЕДЕРАЛЬНОЕ ГОСУДАРСТВЕННОЕ  
БЮДЖЕТНОЕ ОБРАЗОВАТЕЛЬНОЕ УЧРЕЖДЕНИЕ  
ВЫСШЕГО ОБРАЗОВАНИЯ  
«ВОРОНЕЖСКИЙ ГОСУДАРСТВЕННЫЙ  
УНИВЕРСИТЕТ»

А.А. Каменский, А.А. Некипелов, В.В. Чернушкин

**КОМПЬЮТЕРНЫЕ ЛАБОРАТОРНЫЕ ЗАНЯТИЯ  
ПО ТЕОРЕТИЧЕСКОЙ МЕХАНИКЕ  
Часть 2**

Учебное пособие

Воронеж  
Издательский дом ВГУ  
2016

# Содержание

	Предисловие . . . . .	4
7.	Неинерциальные системы отсчета . . . . .	5
	7.1. Движение в равномерно вращающейся системе отсчета . . . . .	5
8.	Малые колебания . . . . .	9
	8.1. Собственные частоты колебаний . . . . .	9
	8.2. Вынужденные колебания . . . . .	16
9.	Формализм Гамильтона . . . . .	20
	9.1. Функция Гамильтона и канонические уравнения . . . . .	20
	9.2. Скобки Пуассона . . . . .	24
	9.3. Канонические преобразования . . . . .	26
	9.4. Метод Гамильтона-Якоби . . . . .	30
10.	Гидродинамика . . . . .	32
	10.1. Идеальная жидкость . . . . .	32
	10.2. Вязкая жидкость . . . . .	35
11.	Функции стандартных библиотек Mathematica . . . . .	37
	11.1. Аналитические преобразования алгебраических выражений . . . . .	38
	11.2. Операции с векторами и матрицами . . . . .	42
	11.3. Дифференцирование и интегрирование. Разложение в ряд Тейлора . . . . .	44
	11.4. Решение дифференциальных уравнений . . . . .	45
	11.5. Построение графиков функций . . . . .	47
	Предметный указатель . . . . .	48

Для решения дифференциального уравнения (при  $E = 0$ ) укажем заранее, что величины  $\sin \alpha$ ,  $\cos \alpha$  и  $\omega$  положительны:

```
--> eq:subst([E=0,v='diff(s,t)],%);
--> assume(sin(alpha)>0 and omega>0);
--> ode2(eq, s, t);
```

Ответив на вопрос Maxima про  $g$ : **positive**, получим ответ:

$$t = \frac{1}{\omega \sin \alpha} \ln \left( 2 \omega \sin \alpha \sqrt{s^2 \omega^2 \sin^2 \alpha - 2 s g \cos \alpha + 2 s \omega^2 \sin^2 \alpha - 2 g \cos \alpha} \right) + const.$$

### Задача 7.2

Определить отклонение от начальной плоскости движения для тела, брошенного с начальной скоростью  $\mathbf{v}_0$

Указание: считать ускорение свободного падения равным  $\mathbf{g} = \{0, 0, -g\}$ , а угловую скорость вращения системы отсчета равной  $\boldsymbol{\Omega} = \{-\Omega \cos \theta, 0, \Omega \sin \theta\}$ , где  $\theta$  – широта.

Решение.

```
--> load(vect);
--> v:[vx,vy,vz]; g1:[0,0,-g];
--> Omega1:[-Omega*cos(theta),0,Omega*sin(theta)];
--> depends([vx,vy,vz],t);
```

Уравнение движения частицы в равномерно вращающейся системе отсчета есть

$$m \frac{d\mathbf{v}}{dt} = m\mathbf{g} + 2m[\mathbf{v} \times \boldsymbol{\Omega}] + m[\boldsymbol{\Omega} \times [\mathbf{r} \times \boldsymbol{\Omega}]]. \quad (7.1)$$

Введем его в координатной форме, пренебрегая слагаемым, квадратичным по  $\boldsymbol{\Omega}$ :

```
--> eq[1]:m*diff(v[1],t)=m*g1[1]+2*m*express(v~Omega1)[1];
--> eq[2]:m*diff(v[2],t)=m*g1[2]+2*m*express(v~Omega1)[2];
--> eq[3]:m*diff(v[3],t)=m*g1[3]+2*m*express(v~Omega1)[3];
```

Подставим  $dv_x/dt$  и  $dv_z/dt$  из первого и третьего уравнений во второе, продифференцировав каждую его часть по времени.

```
--> eliminate([eq[1],diff(eq[2],t),eq[3]],
[diff(vx,t),diff(vz,t)]);
--> eqv:trigsimp(%[1])=0;
```

Теперь дифференциальное уравнение содержит единственную проекцию скорости и допускает понижение степени (на запрос Maxima о знаке параметра введем nonzero):

```
--> ode2(eqv,vy,t);
```

Рассматривая отклонение от начальной плоскости движения, положим в начальный момент времени  $y = 0$ ,  $v_y = 0$ . Остальные начальные условия введем с учетом связи проекций скорости между собой:

```
--> ic2(%,t=0,vy=0,diff(vy,t)=2*(-Omega*cos(theta)*v0z-Omega*sin(theta)*v0x));
```

Поскольку данное приближение справедливо для малых величин  $\Omega$ , то исключим слагаемые избыточной малости, убрав многоточие из ряда Тейлора командой **ratdisrep**

```
--> ratdisrep(ta);
--> r:ratdisrep(taylor(%,Omega,0,1));
```

$$v_y = [(gt^2 - 2v_{0z}t) \cos(\theta) - 2v_{0x}t \sin(\theta)] \Omega.$$

```
--> ode2('diff(y,t)=part(r,2),y,t);
--> ic1(%,t=0,y=0);
--> r:factor(%)
```

Ответ:

$$y = -\Omega t^2 \left[ v_{0z} \cos \theta + v_{0x} \sin \theta - \frac{gt}{3} \right].$$

Замечание: при нулевой начальной скорости (например, тело падает в шахту), получаем кубическую зависимость от времени:

$$y = \frac{g\Omega t^3}{3} \cos(\theta).$$

### Задача 7.3

Определить влияние, оказываемое вращением Земли на малые колебания математического маятника (маятник Фуко).

Решение.

Потенциальная энергия математического маятника равна  $U = mgh$ , причем

$$x^2 + y^2 + z^2 = l^2,$$

где  $l$  – длина нити.

```
--> z:sqrt(l^2-x^2-y^2);
--> U:m*g*(1-z);
```

В дальнейшем потребуется разложение в ряды в символьном виде, поэтому свяжем производные координат со скоростями с помощью команды **gradef**:

```
--> gradef(x,t,vx); gradef(y,t,vy);
--> load(vect);
--> v:[diff(x,t),diff(y,t),diff(z,t)];
--> Omega:[Omegax,Omegay,Omegaz];
```

Сила, действующая на материальную точку маятника равна

$$\mathbf{F} = -gradU :$$

```
--> F:-ev(express(grad(U)),diff);
```

Подставим ее в уравнение движения (7.1), снова пренебрегая слагаемым, квадратичным по  $\Omega$ . Запишем уравнение в координатной форме, ограничившись членами первого порядка малости:

```
--> m*'diff(v[1],t)=F[1]+2*m*express(v~Omega)[1];
--> eqx:ratdisrep(taylor(%, [x,y,vx,vy],0,1));
--> m*'diff(v[2],t)=F[2]+2*m*express(v~Omega)[2];
--> eqy:ratdisrep(taylor(%, [x,y,vx,vy],0,1));
```

Вернувшись к записи через производные, получим систему уравнений в виде

```
--> eq1:expand(subst([x=X(t),y=Y(t),
vx='diff(X(t),t),vy='diff(Y(t),t)],eqx/m));
--> eq2:expand(subst([x=X(t),y=Y(t),
vx='diff(X(t),t),vy='diff(Y(t),t)],eqy/m));
```

$$\left\{ \ddot{X}(t) = 2\Omega_z \dot{Y}(t) - \frac{g}{l} X(t), \quad \ddot{Y}(t) = -2\Omega_z \dot{X}(t) - \frac{g}{l} Y(t) \right\}.$$

Два уравнения легко сводятся к одному с помощью комплексной замены

$$\xi(t) = X(t) + iY(t),$$

для этого умножим (почленно) второе уравнение на мнимую единицу и сложим с первым:

```
--> subst(X(t)=xi(t)-%i*Y(t),eq1+%i*eq2);
--> ev(%,diff);
--> eq:expand(%) ;
```

$$\ddot{\xi}(t) = -2i\Omega_z \dot{\xi}(t) - \frac{g}{l} \xi(t).$$

```
--> ode2(eq,xi(t),t);
```

Отвечив positive на вопрос Maxima о знаке величины  $l(l\Omega_z^2 + g)$ , получим решение в виде

$$\xi(t) = [k_1 \sin(\omega t) + k_2 \cos(\omega t)] e^{-i\Omega_z t},$$

где

$$\omega = \sqrt{\Omega_z^2 + \frac{g}{l}} \approx \sqrt{\frac{g}{l}}.$$

Множитель  $e^{-i\Omega_z t}$  означает медленное вращение плоскости колебаний маятника.

## 8. Малые колебания

### 8.1. Собственные частоты колебаний

#### Задача 8.1

Закон гармонических колебаний задан в виде

$$x(t) = A \sin(\beta t) (\cos^2(\beta t) - \sin^2(\beta t)/3).$$

Найти положение равновесия, частоту и амплитуду колебаний.

Решение.

Для гармонических колебаний со смещенным положением равновесия  $x_{eq}$  можем записать

$$\frac{d^2 x}{dt^2} = -\omega^2 (x(t) - x_{eq}), \quad \frac{d^3 x}{dt^3} = -\omega^2 \frac{dx}{dt}.$$

Отсюда найдем частоту колебаний:

```
--> x:A*sin(beta*t)*(cos(beta*t)^2-sin(beta*t)^2/3);
--> factor(diff(x,t,3)/diff(x,t,1));
--> omega:sqrt(-%);
```

а затем и положение равновесия:

```
--> xeq:trigsimp(x+diff(x,t,2)/omega^2);
```

Амплитуду найдем по формуле

$$a = \sqrt{(x(t) - x_{eq})^2 + (\dot{x}(t)/\omega)^2} : \quad (8.1)$$

```
--> a:sqrt(trigsimp((x-xeq)^2+diff(x,t)^2/omega^2));
```

Ответ:

$$\omega = 3|\beta|, \quad x_{eq} = 0, \quad a = |A|/3.$$

### Задача 8.2

Решить предыдущую задачу для следующих функций  $x(t)$ :

а)  $x(t) = A \sin(2\beta t) + B \sin^2(\beta t)$ ;

б)  $x(t) = A \cos(\beta t + \delta_1) + B \cos(\beta t + \delta_2)$ .

Указание: для автоматического разложения тригонометрических функций кратных аргументов использовать опцию **trigexpand** :

`--> trigexpand:true;`

Ответ:

$$\text{а) } \omega = 2|\beta|, \quad x_{eq} = B/2, \quad a = \frac{1}{2}\sqrt{4A^2 + B^2},$$

$$\text{б) } \omega = |\beta|, \quad x_{eq} = 0, \quad a = \sqrt{A^2 + B^2 + 2AB \cos(\delta_2 - \delta_1)}.$$

### Задача 8.3

Частица массы  $m$  совершает одномерное движение в поле

$$U(x) = U_0 \cos(\alpha x) - F_0 x.$$

Получить условие, при котором могут возникнуть малые колебания и найти их частоту ( $U_0$ ,  $\alpha$ ,  $F_0$  – постоянные).

Решение.

Из условия экстремума потенциальной энергии найдем положение равновесия:

`--> x0:part(solve(diff(U(x),x)=0,x)[1],2);`

Предупреждение "solve: using arc-trig functions to get a solution. Some solutions will be lost." означает, что в формуле на экране

$$-a \sin\left(\frac{F_0}{\alpha U_0}\right) / \alpha$$

не учтен знак  $\pm$ . Выберем далее знак таким, чтобы коэффициент колебаний  $k$  получился положительным, это обеспечивается добавлением модуля.

`--> k:subst(x=x0,diff(U(x),x,2));`

`--> omega:sqrt(ratsimp(abs(k))/m);`

Очевидно, результат применим только в случае положительного значения подкоренного выражения.

Ответ:

$$|F_0| < |U_0 \alpha|, \quad \omega = \sqrt{\frac{|\alpha|}{m}} \sqrt{(U_0 \alpha)^2 - F_0^2}.$$